
Semimagic generating functions and constituents

(general form, with cubic data)

Notation:
S:  semimagic squares (all positive values).
s:   normalized squares (symmetry types).
R:   reduced squares (least element is 0).
r:   reduced normalized squares (reduced symmetry types).
gf:  generating function in some form.
gfsum:  generating function as a sum of simple terms.
c:   Cubic (fixed strict upper bound; weak upper bound for reduced).
a:   Affine (fixed magic sum).
We start by recomputing rs from the semimagic count.  From the Latte results we get
the closed Ehrhart g.f. of each flat, which depends on whether we're doing cubic or 
affine.  We also need 
p =  period (or truncated period in affine),
d = degree/dimension,  
RtoSfactor = the rational function that multiplies Rgf to Sgf and rgf to sgf.
enddegree:  The number of terms of the sequences to print out.  

enddegree:=20;

This is for cubic: set up main constants.
d:=5; p:=60;
RtoSfactor:=x^2/(1-x)^2;

This is also for cubic: set up simplex data.
simplexname[1]:="OABC": ehrgf[1]:= 1/((1-x)^3*(1-x^2)) : dimen
[1]:=3:
simplexname[2]:="OEE2": ehrgf[2]:= 1/((1-x)*(1-x^2)*(1-x^3)) : 
dimen[2]:=2:
simplexname[3]:="OAE2": ehrgf[3]:= 1/((1-x)*(1-x^2)^2) : dimen
[3]:=2:
simplexname[4]:="ADE2": ehrgf[4]:= 1/((1-x^2)^3) : dimen[4]:=2:
simplexname[5]:="DE1E2": ehrgf[5]:= 1/((1-x^2)^2*(1-x^3)) : 
dimen[5]:=2:
simplexname[6]:="OCE": ehrgf[6]:= 1/((1-x)^2*(1-x^3)) : dimen
[6]:=2:
simplexname[7]:="BDE1": ehrgf[7]:= 1/((1-x)*(1-x^2)*(1-x^3)) : 
dimen[7]:=2:



simplexname[8]:="ABD": ehrgf[8]:= 1/((1-x)*(1-x^2)^2) : dimen
[8]:=2:
simplexname[9]:="FG1": ehrgf[9]:= 1/((1-x^3)*(1-x^5)) : dimen
[9]:=1:
simplexname[10]:="EF": ehrgf[10]:= 1/((1-x^3)^2) : dimen[10]:=
1:
simplexname[11]:="OG": ehrgf[11]:= 1/((1-x)*(1-x^4)) : dimen
[11]:=1:
simplexname[12]:="FG": ehrgf[12]:= 1/((1-x^3)*(1-x^4)) : dimen
[12]:=1:
simplexname[13]:="AF": ehrgf[13]:= 1/((1-x^2)*(1-x^3)) : dimen
[13]:=1:
simplexname[14]:="DG": ehrgf[14]:= 1/((1-x^2)*(1-x^4)) : dimen
[14]:=1:
simplexname[15]:="DG2": ehrgf[15]:= 1/((1-x^2)*(1-x^5)) : dimen
[15]:=1:
simplexname[16]:="DE": ehrgf[16]:= 1/((1-x^2)*(1-x^3)) : dimen
[16]:=1:
simplexname[17]:="H": ehrgf[17] := 1/(1-x^5) : dimen[17]:=0:
for n from 1 to 17 do print(simplexname[n], dimen[n], ehrgf[n])
; od;



The closed E.g.f. is converted to the open E.g.f.:
for n from 1 to 17 do 
  mu[n]:=(-1)^(dimen[1]-dimen[n]):
od:
mu[14]:=2*mu[14]:
for n from 1 to 17 do  
  openehrgf[n]:=simplify(-(-1)^dimen[n]*subs(x=1/x,ehrgf[n])):
od:
for n from 1 to 17 do 
  rgfterm[n]:=openehrgf[n]: 
od:

Now compute the number of reduced symmetry types:
rgfsum:=sum(mu[nn]*rgfterm[nn],nn=1..17);
rgf:=simplify(rgfsum):



We need the total number of reduced squares, Rgf:
Rgfsum:=72*rgfsum;
Rgf:=simplify(Rgfsum):

Hence Sgf, the g.f. of the number of semimagic squares, equals
Sgfsum:=RtoSfactor*Rgfsum;
Sgf:=simplify(Sgfsum):

The g.f. of the total number of symmetry types, sgf:
sgfsum:=RtoSfactor*rgfsum;
sgf:=simplify(sgfsum):

Generate the series expansions of the g.f.'s.



Expressing the rational function with standard denominator gives an orders-of-
magnitude speedup in the series expansion.  
Standard denominator (1-x^p)^{d+1}.  

pdenom:=(1-x^p):
standenom:=pdenom^(d+1);

G.f. as rational function with standard denominator.
Sgfstandnum:=simplify(numer(Sgf)*simplify(standenom/denom(Sgf))
):
Sgf:=Sgfstandnum/standenom;

G.f. as rational function with standard denominator.
Rgfstandnum:=simplify(numer(Rgf)*standenom/denom(Rgf)):
Rgf:=Rgfstandnum/standenom;



G.f. as rational function with standard denominator.
sgfstandnum:=simplify(numer(sgf)*simplify(standenom/denom(sgf))
): 
sgf:=sgfstandnum/standenom;

G.f. as rational function with standard denominator.
rgfstandnum:=simplify(numer(rgf)*standenom/denom(rgf)):
rgf:=rgfstandnum/standenom;



Expand the series to find the first few values of the number of squares.
Sseries:=series(Sgf,x=0,enddegree+1);

Expand the series to find the first few values of the number of reduced squares.
Rseries:=series(Rgf,x=0,enddegree+1);

Expand the series to find the first few values of the number of symmetry types.
sseries:=series(sgf,x=0,enddegree+1);

Expand the series to find the first few values of the number of reduced symmetry 
types.

rseries:=series(rgf,x=0,enddegree+1);

Find the constituents

Casculate the zeroth constituent of the total semimagic counting function.  Find its 
constant term.

Szeroth:=expand(
sum(coeff(Sgfstandnum,x,p*jj)*binomial(d+t/p-jj,d),jj=0..d+1) )
;
print(subs(t=0,Szeroth)):  



Extract the constituents of the total semimagic counting function.
Sconstituent[0]:=Szeroth:
for r from 1 to p do
  Sconstituent[r]:=expand(sum( coeff(Sgfstandnum,x,p*jj+r)*
binomial(d+(t-r)/p-jj,d), jj=0..d)):
#  print(r):
#  print( Sconstituent[r] ):
  print( factor(Sconstituent[r]) ):
od;













Extract the coefficients of the constituents.
for r from 1 to p do
 for coeffdeg from 0 to d do 
  Sc[coeffdeg,r]:=coeff(Sconstituent[r],t,coeffdeg):
  #print( r, Sc[coeffdeg,r] ):
 od:
od:

Print and analyze the constituent coefficients for periods.  First the higher 



coefficients, which ought to be constant.  Print the first coefficient, then any that 
don't repeat the preceding value.

for coeffdeg from 3 to d do
  print("degree", coeffdeg, "coeff", Sc[coeffdeg,1]):
  print(1,Sc[coeffdeg,1]);
  for r from 2 to p do
    stepdifference:=Sc[coeffdeg,r]-Sc[coeffdeg,r-1]:
    if( stepdifference<>0 ) then 
    print(r,Sc[coeffdeg,r],stepdifference):
    fi: 
  od: 
  print("Compared all coefficients of degree", coeffdeg); 
od: 

Next, the constant terms, whose period is expected to be p.  Print all constant terms 
up to the presumed period "stepsize".  Print the difference (at step "stepsize") if they 
are not repeating.
Note that the even terms repeat at step 30 (a period of 15, half the expected period).

stepsize:=30; 
for r from 1 to stepsize do
  print(r, Sc[0,r]); 
od: 
for r from stepsize+1 to p do 
  stepdifference:=Sc[0,r]-Sc[0,r-stepsize]:
  if( stepdifference<>0 ) then print(r,Sc[0,r],stepdifference):
fi:
  #print(r,Sc[0,r],stepdifference);
od:
print("Constant terms completed."); 







Now, the linear terms.  First print all linear coefficients up to the presumed period 
"stepsize"..  Then analyze for period and print the difference (at step "stepsize") if 
they are not repeating.

stepsize:=6; 
for r from 1 to stepsize do
  print(r, Sc[1,r]); 
od: 
for r from stepsize+1 to p do 
  stepdifference:=Sc[1,r]-Sc[1,r-stepsize]:
  if( stepdifference<>0 ) then print(r,Sc[1,r],stepdifference):
fi:
od:
print("Linear coefficients completed."); 



The quadratic terms.  First print all quadratic coefficients up to the presumed period 
"stepsize"..  Then analyze for period and print the difference (at step "stepsize") if 
they are not repeating.

stepsize:=2; 
for r from 1 to stepsize do
  print(r, Sc[2,r]); 
od: 
for r from stepsize+1 to p do 
  stepdifference:=Sc[2,r]-Sc[2,r-stepsize]:
  if( stepdifference<>0 ) then print(r,Sc[2,r],stepdifference):
fi:
od:
print("Quadratic coefficients completed."); 

Casculate the zeroth constituent of the semimagic symmetry-type counting 
function.  Find its constant term.

szeroth:=expand(
sum(coeff(sgfstandnum,x,p*jj)*binomial(d+t/p-jj,d),jj=0..d+1) )
;
print(subs(t=0,szeroth)):  

Extract the constituents of the semimagic symmetry-type counting function.
sconstituent[0]:=szeroth:
for r from 1 to p do
  sconstituent[r]:=expand(sum( coeff(sgfstandnum,x,p*jj+r)*
binomial(d+(t-r)/p-jj,d), jj=0..d)):
#  print(r):
#  print( sconstituent[r] ):
  print( factor(sconstituent[r]) ):
od; 















Extract the coefficients of the constituents.
for r from 1 to p do
 for coeffdeg from 0 to d do 
  sc[coeffdeg,r]:=coeff(sconstituent[r],t,coeffdeg):
  #print( r, sc[coeffdeg,r] ):
 od:
od: 

Print and analyze the constituent coefficients for periods.  First the higher 
coefficients, which are constant.  Print the first coefficient, then any that don't repeat
the preceding value (there are none).

for coeffdeg from 3 to d do
  print("degree", coeffdeg, "coeff", sc[coeffdeg,1]):
  for r from 2 to p do
    stepdifference:=sc[coeffdeg,r]-sc[coeffdeg,r-1]:
    if( stepdifference<>0 ) then 
    print(r,sc[coeffdeg,r],stepdifference):
    fi: 
  od: 
od: 



Next, the constant terms, whose period is expected to be 60.  Print all constant terms
up to the presumed period "stepsize".  Print the difference (at step "stepsize") if they 
are not repeating.

stepsize:=30; 
for r from 1 to stepsize do
  print(r, sc[0,r]); 
od: 
for r from stepsize+1 to p do 
  stepdifference:=sc[0,r]-sc[0,r-stepsize]:
  if( stepdifference<>0 ) then print(r,sc[0,r],stepdifference):
fi:
od:
print("Constant terms completed."); 





Now, the linear terms.  First print all linear coefficients up to the presumed period 
"stepsize".  Then analyze for period and print the difference (at step "stepsize") if 
they are not repeating.

stepsize:=6; 
for r from 1 to stepsize do
  print(r, sc[1,r]); 
od: 



for r from stepsize+1 to p do 
  stepdifference:=sc[1,r]-sc[1,r-stepsize]:
  if( stepdifference<>0 ) then print(r,sc[1,r],stepdifference):
fi:
od:
print("Linear coefficients completed."); 

The quadratic terms.  First print all quadratic coefficients up to the presumed period 
"stepsize".  Then analyze for period and print the difference (at step "stepsize") if 
they are not repeating.

stepsize:=2; 
for r from 1 to stepsize do
  print(r, sc[2,r]); 
od: 
for r from stepsize+1 to p do 
  stepdifference:=sc[2,r]-sc[2,r-stepsize]:
  if( stepdifference<>0 ) then print(r,sc[2,r],stepdifference):
fi:
od:
print("Quadratic coefficients completed.");




